In terms of analysis, it will be most helpful, I think, to work with increasingly complex subsets of the full model. The first submodel is the model without parasites.

In[103]:=
```
Srate = 3 γ (α^(1/3) Linf S^(2/3) - S);
```

In[104]:=
```
dG = Imax S - ρ G
dC = ρ (1 - eA) G - ρ C
dS = Srate
dR = eR (ρ eA G - (m + mc k) (S + R) - eG Srate)
```

Out[104]= $\text{Imax}\ S - G\ \rho$

Out[105]= $-C\ \rho + (1 - eA)\ G\ \rho$

Out[106]= $3\ \left(-S + \text{Linf}\ S^{2/3}\ \alpha^{1/3}\right)\ \gamma$

Out[107]= $eR\ \left(-(m + k\ mc)\ (R + S) - 3\ eG\ \left(-S + \text{Linf}\ S^{2/3}\ \alpha^{1/3}\right)\ \gamma + eA\ G\ \rho\right)$

We can compute S^*, the structural volume at equilibrium, analytically, to be equal to $S^* = \alpha\, L_\infty{}^3$. Given this, what are the other equilibria?

$G^* = \frac{\iota_{\max} S^*}{\rho}$

$C^* = (1 - \epsilon_A) \frac{\iota_{\max} S^*}{\rho}$

$R^* = \left(\frac{\epsilon_A\, \iota_{\max}}{m + k\, m_c} - 1\right) S^*$

Note that this conclusion is unrealistic, as it suggests that, at equilibrium, the ratio of structure to reserves, $R^*/S^* = \frac{\epsilon_A\, \iota_{\max}}{m + k\, m_c} - 1$. In order for this to be at or near 0.2-0.3, $\iota_{\max}$ and m must be of similar magnitude. This makes no biological sense, as they should be more like an order of magnitude different from one another.

We address this problem mathematically by considering condition-dependent ingestion: the animal has a target R/S ratio that it wishes to maintain; when it is below this condition, it accelerates ingestion rate; when it is above this condition, ingestion rate drops. You can incorporate that into the following model:

$$dG = \frac{\text{Imax}\ S^{2/3}}{1 + \text{Exp}[\eta\ (R/S - \theta)]} - \rho\ G$$

```
dC = ρ (1 - eA) G - ρ C
dS = Srate
dR = eR (ρ eA G - (m + mc k) (S + R) - eG Srate)
```

The equilibrium R depends on the equilibrium gut biomass, as above.

In[141]:=
```
Assuming[Linf > 0 && α > 0, Simplify[Solve[dR == 0, R] /. S → α Linf^3]] /. α → S / Linf^3
```

Out[141]= $\left\{\left\{R \to \frac{-(m + k\ mc)\ S + eA\ G\ \rho}{m + k\ mc}\right\}\right\}$

You can use this to find an expression for R/S, but it doesn't really help you much. You still get stuck. Suffice it to say that, at equilibrium, there will be some value of R/S that will be larger than θ, and this will cause the ingestion rate to be slightly smaller than $\iota_{\max}$.

In[152]:=
```
pars = {Imax → 2,
    θ → 1 / 3.93,
    ρ → 2,
    η → 10,
    eA → 0.85,
    eG → 1,
    eR → 1,
    eI → 10^-5,
    Linf → 2.94,
    α → 1,
    ɣ → 0.02,
    m → 0.2,
    mc → 0.03,
    mi → 0.03,
    k → 4.9 × 10^-5};
```

Numerical simulations with the above parameter values produce a mouse that ends up with an $R/S = 0.277777$. Given these parameter values, that means that the effective ingestion rate of such a mouse is:

In[155]:=
```
Imax / (1 + Exp[η (R / S - θ)]) /. pars /. R → 0.277777 S
```

Out[155]=
```
0.883905
```

Because we are considering parasites of the colon, structure and reserves will both settle down to this ratio if the parasite does not provoke an immune response. Let's consider that unrealistic case to help guide the choice of parameter values for the parasite. The ingestion rate is replaced by the effective ingestion rate calculated above:

In[160]:=
```
dG = Ieff S - ρ G;
dC = ρ (1 - eA) G - ρ C - σ C P / (H + C);
dP = eP σ C P / (H + C) - μ P;
Eq1 = Solve[{dG, dC, dP} == 0, {G, C, P}]
```

Out[163]=
$$\left\{\left\{G \to \frac{\text{Ieff}\, S}{\rho},\ C \to \frac{\text{Ieff}\, S - \text{eA}\, \text{Ieff}\, S}{\rho},\ P \to 0\right\},\ \left\{G \to \frac{\text{Ieff}\, S}{\rho},\ C \to -\frac{H \mu}{\mu - \text{eP}\, \sigma},\right.\right.$$
$$\left.\left.P \to \frac{1}{\mu\ (\mu - \text{eP}\, \sigma)}\, \text{eP}\ (\text{Ieff}\, S\, \mu - \text{eA}\, \text{Ieff}\, S\, \mu + H \mu \rho - \text{eP}\, \text{Ieff}\, S\, \sigma + \text{eA}\, \text{eP}\, \text{Ieff}\, S\, \sigma)\right\}\right\}$$

It is clear that the parasite will set the total biomass of the colon. Given what the simulations suggest is a reasonable amount of food in the colon (0.57 g), we want to choose these parameter to make this value make sense. However, we also want to ensure that we have a reasonable biomass of parasites as well. What is reasonable? Well, some old data on infection of rats with tapeworm (Hymenolepis) suggests that a chronic tapeworm infection has worm biomass of ~100 mg (0.1 g). So, not much.

In[268]:=
```
Eq1[[2]] /. {Ieff → 0.27777, S → α Linf^3} /. pars
Pstar = (Eq1[[2]] /. {Ieff → 0.27777, S → α Linf^3} /. pars)[[3]][[2]];
Pstar == eP / μ ((1.0588113524520004`) - 2 (-H μ / (μ - eP σ))) // Simplify
```

Out[268]=
$$\left\{G \to 3.52937,\ C \to -\frac{H \mu}{\mu - \text{eP}\, \sigma},\ P \to \frac{\text{eP}\ (1.05881\, \mu + 2\, H \mu - 1.05881\, \text{eP}\, \sigma)}{\mu\ (\mu - \text{eP}\, \sigma)}\right\}$$

Out[270]=
```
True
```

Let's look a little more closely: you can show that the equilibrium $P^* = \frac{\epsilon_P}{\mu}\ (1.06 - 2\, C^*)$, and $C^* = \frac{H}{\frac{\epsilon_P}{\mu}\, \sigma - 1}$. So what is critical is not either ϵ_P or μ, but rather their ratio. That is useful to know. In particular, if $\mu < \epsilon_P$, which seems reasonable to me, then the only way for P^* to be small is for C^* to be large - near 0.53, which is essentially the colon biomass

in the absence of infection.

Considering that we want C^* to be fairly low - the parasite should eat most of what is there in the absence of any immune regulation - then then ϵ_P/μ should be small, which would require $\mu > \epsilon_P$. This is one of those mathematical realities that feels biologically wrong.

Maybe C^* should not be so small. Let's imagine choosing parameters such that C^* is half what it would be normally, so about 0.25. Then to get $P^* = 0.1$, you would need $\epsilon_P/\mu = \frac{0.1}{1.06-2\,(0.25)} = 0.18$, that is $\epsilon_P = 0.18\,\mu$. To get $C^* = 0.25$, I need to choose H and σ such that $0.25(0.15\,\sigma - 1) = H$.

Here is a set of parameters that seems to produce something reasonable.

In[281]:=
```
Eq1[[2]] /. {Ieff → 0.27777, S → α Linf^3} /. pars /. {eP → 0.15, μ → 1, σ → 10, H → 0.1}
```

Out[281]=
```
{G → 3.52937, C → 0.2, P → 0.0988217}
```

In[291]:=
```
pars = {Imax → 2, θ → 0.2544529262086514`, ρ → 2, η → 10, eA → 0.85`,
  eG → 1, eR → 1, eI → 1/100 000, Linf → 2.94`, α → 1, γ → 0.02`, m → 0.2`,
  mc → 0.03`, mi → 0.03`, k → 0.000049, eP → 0.15, μ → 1, σ → 10, H → 0.1};
```

Okay, the next step (really, the final step analytically) is to look at what happens when you add only a constitutive immune response. Again, this has the advantage of not affecting either *S* or *R*. All that we need to do is add in the effect of the immune response on the parasite.

In[304]:=
```
dG = Ieff S - ρ G;
dC = ρ (1 - eA) G - ρ C - σ C P / (H + C);
dP = eP σ C P / (H + C) - μ P - μc P Ii;
Eq2 = Solve[{dG, dC, dP} == 0, {G, C, P}]
```

Out[307]=
```
{{G → Ieff S/ρ, C → (Ieff S - eA Ieff S)/ρ, P → 0},
 {G → Ieff S/ρ, C → -(H (μ + Ii μc))/(μ + Ii μc - eP σ),
  P → (eP (Ieff S μ - eA Ieff S μ + Ieff Ii S μc - eA Ieff Ii S μc +
      H μ ρ + H Ii μc ρ - eP Ieff S σ + eA eP Ieff S σ)) / ((μ + Ii μc) (μ + Ii μc - eP σ))}}
```

Okay, let's look at this a bit more carefully. In order for C^* to be positive, it must be the case that $I_i\,\mu_i < 0.5$.

```
Eq2[[2]] /. pars
```

Out[308]=
```
{G → Ieff S/2, C → -(0.1 (1 + Ii μc))/(-0.5 + Ii μc),
 P → (0.15 (0.2 - 0.075 Ieff S + 0.2 Ii μc + 0.15 Ieff Ii S μc)) / ((-0.5 + Ii μc) (1 + Ii μc))}
```

But actually, for an adult, I know exactly how big I_i is, because $I_i = k(S + R)$; a healthy mouse, according to the simulation, has a total biomass of ~32g. Based on Sarah's mouse data, k = 4.9e-5, so $I_i = 0.0016$. Here is a range of μ_c values that produce C^* values between the value induced by parasites on their own and the parasite-free equilibrium, and P^* values less than the value attained by the parasite on its own. So, a $\mu_c = 100$ seems to work fairly well.

In[316]:=
```
Table[Eq2[[2]] /. {Ieff → 0.27777, S → α Linf^3} /. pars /. {Ii → 0.0016}, {μc, 10, 150, 10}]
```

Out[316]=
```
{{G → 3.52937, C → 0.209917, P → 0.0943371},
 {G → 3.52937, C → 0.220513, P → 0.0897944}, {G → 3.52937, C → 0.231858, P → 0.0851757},
 {G → 3.52937, C → 0.244037, P → 0.0804612}, {G → 3.52937, C → 0.257143, P → 0.0756286},
 {G → 3.52937, C → 0.271287, P → 0.0706529}, {G → 3.52937, C → 0.286598, P → 0.0655057},
 {G → 3.52937, C → 0.303226, P → 0.0601542}, {G → 3.52937, C → 0.321348, P → 0.0545605},
 {G → 3.52937, C → 0.341176, P → 0.04868}, {G → 3.52937, C → 0.362963, P → 0.0424599},
 {G → 3.52937, C → 0.387013, P → 0.0358371}, {G → 3.52937, C → 0.413699, P → 0.0287352},
 {G → 3.52937, C → 0.443478, P → 0.0210606}, {G → 3.52937, C → 0.476923, P → 0.0126974}}
```